
1 はじめに

本稿は、[2]の 5.5節 (命題信念様相論理)への補足を意

図して書かれている。具体的には、その節で定理 5.5とし

て証明なしに述べられている、命題信念様相論理 (命題様

相論理体系 KD45に相当)の Hilbert流の演繹体系1の完全

性に対し、証明を与えることを目的とする。

Hilbert流の演繹体系については、完全性の証明も、ま

た、与えられた論理式の証明を作り出すことも、直感的

には行いにくい。一方、2.3節で述べるシーケント計算と

呼ばれる演繹体系は、特に命題論理/命題様相論理につい

ては、完全性の証明は、極めて直感に沿った形で行える2。

それは、論理式 Aが証明可能でないと仮定すると、自然

な形で、Aを偽にする真理値割り当て (命題論理の場合)、

ないしはクリプキ ·モデル (命題様相論理の場合)が作れ

ることによる。また、それと同じ考え方で、論理式の証

明を作り出すアルゴリズムの存在も自然に示せる [1]。

そこで本稿では、シーケント計算による証明を、Hilbert

流の演繹体系での演繹に変換することによって、命題論理

や命題様相論理K, KD45のHilbert流の演繹体系における

完全性の証明や、同体系での論理式の証明を作り出す方

法が得られることを示すという方針をとる。特に、Hilbert

流での命題様相論理 KD45の演繹体系の完全性の証明は、

各種書籍にも意外に取り上げられていないので、それを

補うことをも目的とする。

以下、2節では命題 (非様相)論理、3節では命題様相

論理 K の体系について述べ、次いで 4節で命題様相論理

KD45について述べる。

以後、A, B, · · · は論理式、Γ, ∆, · · · は論理式の multi

setを表す。また、古典 (非様相)論理記号は ⊃, ¬のみと
し、A ∧ B は ¬(A ⊃ ¬B)の、A ∨ B は ¬A ⊃ B の、そ

れぞれ略記とする。論理記号の優先順位は、単項演算子 ·
∧ · ∨ ·⊃の順に高いとする。⊃は右結合とするが、しばし
ば、紛れにくいよう A ⊃ B ⊃ C のような場合も明示的

1「Hilbert流」の演繹体系とは、[2] の 2.5節 (命題論理の公理と推
論規則) や 5.1.5項 (体系 K)、5.5.4項 (命題信念様相論理の体系) のよ
うに、公理と推論規則から定理となる論理式を演繹する体系のことを指
す。なお、[2] では「演繹体系」を「公理系」と呼んでいるが、本稿で
は「演繹体系」で統一する。

2何をもって「直感的」「直感に沿った」と呼ぶかには個人差もあろ
うが、少なくとも筆者には、Hilbert流よりもシーケント計算に対する
完全性の証明の方がずっと直感的でわかりやすいものと感じる。

に括弧をつけてA ⊃ (B ⊃ C)と書く。その他の二項演算

子は左結合とする。

2 命題論理

2.1 Hilbert流の演繹体系

命題 (非様相)論理の Hilbert流の演繹体系を導入する。

2.1.1 公理

「:」の左はそれぞれの公理の名前である。

Ax1 : A ⊃ (B ⊃ A)

Ax2 : (A ⊃ (B ⊃ C)) ⊃ ((A ⊃ B) ⊃ (A ⊃ C))

Ax3 : (¬A ⊃ ¬B) ⊃ ((¬A ⊃ B) ⊃ A)

2.1.2 規則

規則の横線の上の論理式の集合をその規則の「前提」、

下の論理式を「結論」と呼ぶ。横線の右はその規則の名

前である。

A A ⊃ B
B

MP

2.1.3 演繹

公理と Γから規則を使ってAが導けることを Γ ⊢ Aと

書く。正確に言うと、Γ ⊢ Aとは、論理式をノードとし、

かつ以下を満たす有限木が存在することである。

1. 根は A

2. 葉は Γの要素あるいは公理

3. 葉でない任意のノード N に対し、N を結論とし、

N の子ノードの集合を前提とする規則が存在する

また、その木を Γからの Aの「演繹」、あるいは紛れが

なければ単に「演繹」と呼ぶ。特に、∅からの Aの演繹

を Aの「証明」と呼ぶ。

なお、今のところ規則はMPだけである。

1



しばしば、論理式の multi setと論理式の列を適切に同

一視する。すなわち、{A1, · · · , An} ⊢ AのことをA1, · · · ,
An ⊢ A と (特に、∅ ⊢ A のことを ⊢ A と) 書いたり、

Γ ∪ {A1, · · · , An} ⊢ B のことを Γ, A1, · · · , An ⊢ B と書

いたりする。

⊢ Aのとき Aを定理という。

2.2 演繹定理

命題論理の Hilbert流の演繹体系に対する、演繹定理を

示す。

2.2.1 補助規則その 1

以下は、本来の規則と同様、規則として用いることが

できる。これらを以降では補助規則と呼ぶ。横線の右側

はその補助規則の名前である。

A
B ⊃ A

Weak
A ⊃ B A ⊃ (B ⊃ C)

A ⊃ C
MP2

A ⊃ B B ⊃ C
A ⊃ C

MP1

上 2つは公理と MP からすぐ、MP1 は上 2つから得ら

れる。

2.2.2 補助定理その 1

定理のうち、公理と同様に用いるものを以降では補助

定理と呼ぶ。次のものは補助定理である (「:」の左は補助

定理の名前)。この式は後に演繹定理の証明で使われるこ

とに注意。

ID : A ⊃ A

証明は以下から得られる (B は任意でよい)。

A ⊃ (B ⊃ A) A ⊃ ((B ⊃ A) ⊃ A)

A ⊃ A
MP2

2.2.3 演繹定理

定理 2.1 Γ ⊢ A ⊃ B iff Γ, A ⊢ B

証明 左から右へは MP があるので明らか。右から左へ

は、Γ, AからBへの演繹の木のノード数に関する帰納法

による。

• 木のノードが 1つだけなら、そのノードBはΓの要

素か公理かAである。B = Aの場合はA ⊃ Aが証

明でき、それ以外の場合は BからWeakで A ⊃ B

を導けるため、いずれの場合も Γから A ⊃ Bへの

演繹が作れる。

• 木のノードが 2つ以上の場合、木の根はBである。

規則はMPだけであるため、ある論理式B′が存在し

て、Γ, AからB′への演繹の木と、Γ, AからB′ ⊃ B

への演繹の木があり、Bの子ノードはB′とB′ ⊃ B

である。帰納法の仮定から、Γから A ⊃ B′への演

繹の木と、Γから A ⊃ (B′ ⊃ B)への演繹の木が存

在する。それらの根から MP2で A ⊃ B を導ける

ことから、Γから A ⊃ B への演繹も存在する。

証明には、規則がMPだけであることと、IDが定理で

あることが使われている。従って、MP以外の規則 (補助

規則は除く) を持つ演繹体系では、定理 2.1の右から左

は必ずしも成り立たない (例えば 3節の命題様相論理で、

A ⊢ □Aだからといって ⊢ A ⊃ □Aとはならない)。ま

た、公理が少なくて IDを証明できない場合も、右から左

は必ずしも成り立たない。

この証明は、「Γ, Aからの B の演繹」を与えて「Γか

らのA ⊃ Bの演繹」を作るアルゴリズムの存在も示して

いる。

2.3 シーケント計算

以下、シーケント計算による演繹体系を導入する。本

稿でのシーケント計算の定義は基本的に [1]のものを踏襲

する。

A1, · · · , Am → B1, · · · , Bnの形をしたもの (但し 0 ≤ m,

0 ≤ nで、A1, · · · , Am, B1, · · · , Bnは論理式)をシーケン

トという。シーケントの「→」の左のみ、または右のみの
順番を入れ替えたものは、元のシーケントと同一視する。

2.3.1 イニシャルシーケント

「→」の左と右に同一の論理式があるシーケントをイ
ニシャルシーケントという。

2.3.2 シーケント計算の推論規則

シーケント計算の各推論規則は、横線の上にシーケン

トが 0個以上、下にシーケントが 1個という形をしてい

る。線の上のシーケントをその規則の「前提」、下のシー

ケントを「結論」と呼ぶ。

命題論理のシーケント計算の推論規則は以下のもので

ある。論理記号が⊃と ¬だけなので、これらに対する規
則しかない。

Γ → ∆, A Γ, B → ∆

Γ, A ⊃ B → ∆
⊃左

Γ, A → ∆, B

Γ → ∆, A ⊃ B
⊃右

Γ → ∆, A

Γ,¬A → ∆
¬左

Γ, A → ∆

Γ → ∆,¬A ¬右

いずれも、結論のシーケントは「→」の左か右に⊃ある
いは ¬があり、前提のシーケントではその⊃あるいは ¬
が外れる形となっている。さらに、「→」の左か右に⊃か
¬が 1つ以上ある任意のシーケント S に対し、S を結論

とする推論規則が必ず存在する。これらの性質は 2.3.3節

で使う。

証明可能なシーケントを以下のように定義する。

• イニシャルシーケントは証明可能

• 推論規則の前提のシーケントが全て証明可能ならば
結論のシーケントも証明可能

また、論理式Aが証明可能であるとは、シーケント→ A

が証明可能であることとする。

2.3.3 シーケント計算の完全性その他

シーケント計算の健全性 ·完全性は容易に示せる。また、
シーケントを (あるいは論理式を)与えて、それが証明可

能ならその証明を実際に与え、さもなければ証明可能で

ないと答えるアルゴリズムの存在も容易に知れる。詳細

は [1] を参照。以下に概略を示す。

(命題論理の意味論が定義済みであるとして)シーケン

ト A1, · · · , Am → B1, · · · , Bnの真理値を、論理式 ¬A1 ∨
· · ·∨¬Am∨B1∨ · · ·∨Bnの真理値と同じと定義し、また

この論理式が恒真のとき、元のシーケントも恒真という。

すると、イニシャルシーケントは恒真であり、また、シー

ケント計算の推論規則の前提が全て恒真ならば結論も恒

真である。これより、証明可能なシーケントは恒真であ

り、よって証明可能な論理式も恒真である。

一方、恒真なシーケントが証明可能であることを示す

には、シーケント内の論理オペレータ (⊃, ¬)の個数 nに

関する帰納法による。n = 0でかつ恒真なシーケントは

イニシャルシーケントであるので、証明可能である。次

に n ≤ kの場合を仮定し、n = k+1でかつ恒真なシーケ

ント Sがあるとすると、Sの→の左か右に論理オペレー
タ (⊃, ¬)が少なくとも 1つあるため、S を結論とする推

論規則が必ず存在する。その推論規則の前提はいずれも

恒真であり (各推論規則につき、結論が恒真なら前提も恒

真であることが示せるため)、かつ論理オペレータは k個

以下なので、帰納法の仮定から証明可能である。従って

S も証明可能である。恒真な論理式が証明可能であるこ

とは、このことからすぐ従う。

また、任意に与えたシーケント S が証明可能か否かは

以下のようにして求まる。S 内に論理オペレータ (⊃, ¬)

が 1つ以上あれば、Sを結論とする推論規則が存在し、そ

の前提はいずれも論理オペレータの個数が Sより少ない。

それらのシーケントに対し同様の過程を繰り返すと、論

理オペレータの個数が減っていき、いずれは 0個になる。

そのようにして得られた論理オペレータ 0個のシーケン

トが全てイニシャルシーケントであれば S は証明可能で

あり、またイニシャルシーケントでないものが 1つでも

あればそのシーケントは恒真ではないので、S も恒真で

はなく、従って証明可能ではない。

2.4 Hilbert流の演繹体系の完全性

まず、Hilbert流の演繹体系での追加の補助規則をいく

つか導入し、続いて、シーケント計算での証明からHilbert

流演繹への変換を示すことにより、Hilbert流の演繹体系

の完全性を示す。

2.4.1 Hilbert流の演繹体系での補助規則その 2

¬A ⊃ ¬B ¬A ⊃ B
A

PD1
¬A ⊃ ¬B B

A
PD2

¬B B
A

IN
¬¬A
A

DN
A ⊃ ¬B A ⊃ B

¬A PD3

PD1は公理とMPから、PD2と INの2つはPD1とWeak

からすぐ。DNは下記から得られる。

¬¬A
¬A ⊃ ¬¬A Weak ID

¬A ⊃ ¬A
A

PD1

演繹定理により、INおよびDNは補助定理¬B ⊃ (B ⊃ A),

¬¬A ⊃ Aとしても使える。このことから、PD3も以下

のようにして得られる。

補助定理 (DN)
¬¬A ⊃ A A ⊃ ¬B

¬¬A ⊃ ¬B MP1

補助定理 (DN)
¬¬A ⊃ A A ⊃ B

¬¬A ⊃ B
MP1

¬A PD1

2.4.2 シーケント計算での証明から Hilbert 流演繹への

変換

以下、Γの論理式の1つ1つに¬を付けたものを¬Γと書
く (例えば Γ = {p, p ⊃ q}ならば ¬Γ = {¬p,¬(p ⊃ q)})。

Hilbert流の公理のうち適当に選んだ 1つを T とする。

シーケント S = Γ → ∆に対し、Hilbert流の演繹体系で

の Γ,¬∆から ¬T への演繹が存在するならば、それを「S

の Hilbert流演繹」と呼ぶことにする。

定理 2.2 任意の証明可能なシーケント S に対し、S の

Hilbert流演繹が存在する。

証明 イニシャルシーケントの Hilbert流演繹が存在する

こと、および、シーケント計算の各推論規則について、前

提のシーケントの全てに Hilbert流演繹が存在するならば

結論のシーケントにも Hilbert流演繹が存在することを示

せばよい。

• S がイニシャルシーケント Γ, A → ∆, Aの場合、S

の Hilbert流演繹は以下のように作れる (Γ, A,¬∆,

¬A ⊢ ¬T )。

A ¬A
¬T IN

• ⊃左の前提それぞれにHilbert流演繹が存在する (Γ,

¬∆,¬A ⊢ ¬T および Γ, B,¬∆ ⊢ ¬T )場合、演繹定

理より Γ,¬∆ ⊢ ¬A ⊃ ¬T および Γ,¬∆ ⊢ B ⊃ ¬T
であり、以下により結論の Hilbert流演繹を作れる

(Γ, A ⊃ B,¬∆ ⊢ ¬T )。T が公理であることに注意。

¬A ⊃ ¬T T
A

PD2
A ⊃ B

B
MP

B ⊃ ¬T
¬T MP

• ⊃右の前提に Hilbert流演繹が存在する (Γ, A,¬∆,

¬B ⊢ ¬T )場合、以下により結論の Hilbert流演繹

を作れる (Γ,¬∆,¬(A ⊃ B) ⊢ ¬T )。

¬(A ⊃ B)

¬A ⊃ ¬(A ⊃ B)
Weak

補助定理 (IN)
¬A ⊃ (A ⊃ B)

A
PD1

¬(A ⊃ B)

B ⊃ ¬(A ⊃ B)
Weak

B ⊃ (A ⊃ B)

¬B PD3

• ¬左の前提にHilbert流演繹が存在する場合、結論の

Hilbert流演繹はそれと同じである。

• ¬右の前提に Hilbert流演繹が存在する (Γ, A,¬∆ ⊢
¬T )場合、DNにより結論の Hilbert流演繹を作れ

る (Γ,¬∆,¬¬A ⊢ ¬T )。

また、以上の証明からは、シーケント S の証明が与え

られていれば、Sの Hilbert流演繹を具体的に作れること

もわかる。

系 2.3 命題 (非様相)論理のHilbert流の演繹体系は健全か

つ完全。

証明 健全性は容易。完全性を示すため、論理式Aが恒真

とする。シーケント計算の完全性より、シーケント→ A

は証明可能である。よって、→ AのHilbert流演繹が存在

する。すなわち ¬A ⊢ ¬T である。演繹定理と PD2、お

よび T が公理であることから ⊢ Aとなる。

以上からはさらに、恒真な論理式Aに対し、AのHilbert

流体系での証明を具体的に作れることもわかる。

3 命題様相論理K

3.1 Hilbert流の演繹体系

命題様相論理 KのHilbert流の演繹体系を導入する。以

下、Γの論理式の 1つ 1つに□を付けたものを□Γと書く

(例えば Γ = {p, p ⊃ q}ならば □Γ = {□ p,□(p ⊃ q)})。

3.1.1 公理

命題論理の公理に以下を加えたもの。

K : □(A ⊃ B) ⊃ (□A ⊃ □B)

正確に言うと、命題論理の公理のA, B, Cを命題様相論理

の論理式で置き換えたものも公理 (例えば□A ⊃ (□B ⊃
□A)がそう)である。

3.1.2 規則

命題論理の規則に以下を加えたもの。

A
□A

Nec

演繹の定義は 2.1.3節と同じ。

以降では、演繹のうち、途中の過程で「定理でない論理

式 Aに対し Necを適用」していないもの (すなわち、演

繹の木に□Aの唯一の子ノードAでかつ定理でないもの

が存在しないもの)を「正規演繹」と呼ぶことにし、Γか

ら Aへの正規演繹があるとき Γ ⊢◦ Aと書く。

3.1.3 演繹定理

演繹定理は制限付きでしか成り立たない。

定理 3.1 Γ ⊢◦ A ⊃ B iff Γ, A ⊢◦ B

証明 「演繹」が「正規演繹」に変わること以外、定理 2.1

の証明とほぼ同様。特に、規則 Necが加わったため、場

合分けの中に、木の根 B が □B′ の形をしていて、B の

唯一の子ノードがB′である場合が加わる。この場合 ⊢◦ B
なので、Weakにより ⊢◦ A ⊃ Bでもある。従って Γから

A ⊃ B への正規演繹が存在する。

3.2 シーケント計算

命題様相論理 K のシーケント計算による演繹体系を導

入する。シーケント、イニシャルシーケントの定義は 2.3

節と同じ。

3.2.1 シーケント計算の推論規則

命題論理のシーケント計算の推論規則に以下を加えた

もの。

Γ → A
□Γ → □A

Nec
Γ → ∆

Γ,Γ′ → ∆,∆′ Weak

シーケントや論理式が証明可能であることの定義も 2.3.2

節と同じ。

体系 K に関しても、シーケント計算の健全性や完全性

は、命題論理のシーケント計算よりは少し難しくなるも

のの比較的容易に示せる。また、シーケントを (あるいは

論理式を)与えて、それが証明可能ならその証明を実際に

与え、さもなければ証明可能でないと答えるアルゴリズ

ムの存在も示せる。これらについても [1] 参照。

3.3 体系K でのHilbert 流の演繹体系の完全性

Hilbert流の公理のうち適当に選んだ 1つを T とする。

シーケント S = Γ → ∆に対し、Hilbert流の演繹体系で

の Γ,¬∆から ¬T への正規演繹が存在するならば、それ
を「S の Hilbert流正規演繹」と呼ぶことにする。

定理 3.2 任意の証明可能なシーケント S に対し、S の

Hilbert流正規演繹が存在する。

証明 定理 2.2とほぼ同じ。特に、

• Weakの前提に Hilbert流正規演繹が存在する場合、

結論にHilbert流正規演繹が存在することは明らか。

• Necの前提にHilbert流正規演繹が存在する (A1, · · · ,
An,¬A ⊢◦ ¬T )場合、演繹定理および PD2よりA1,

· · · , An ⊢◦ Aとなり、さらに演繹定理から ⊢◦ A1 ⊃
A2 ⊃ · · · ⊃ An ⊃ A、Necから ⊢◦ □(A1 ⊃ A2 ⊃
· · · ⊃ An ⊃ A)を得る。

ここから公理 K と MP を用いて □A1 ⊢◦ □(A2 ⊃
· · · ⊃ An ⊃ A)を得、さらに同様の過程を繰り返

すことで □A1, · · · ,□An ⊢◦ □A を得る。IN から

□A1, · · · ,□An,¬□A ⊢◦ ¬T となり、従って Nec

の結論にも Hilbert流正規演繹が存在する。

また、以上から、シーケント S の証明が与えられてい

れば、Sの Hilbert流正規演繹を具体的に作れることもわ

かる。

系 3.3 K の Hilbert流の演繹体系は健全かつ完全。

証明 系 2.3と同様。特に、論理式Aが恒真の場合、シー

ケント→ AのHilbert流正規演繹が存在することに注意。

また、恒真な論理式Aに対し、AのHilbert流体系での

証明を具体的に作れることもわかる。

4 命題様相論理KD45

4.1 Hilbert流の演繹体系

命題様相論理 KD45の Hilbert流の演繹体系は、命題様

相論理 K の公理に以下を加えたものである。

D : □A ⊃ ¬□¬A

4 : □A ⊃ □□A

5 : ¬□A ⊃ □¬□A

規則は命題様相論理 Kと同じ。演繹の定義も Kと同じで

ある。演繹定理は定理 3.1と同じ形で成り立つ。

4.2 シーケント計算

命題様相論理 KD45のシーケント計算による演繹体系

を導入する。特に、KD45での完全性に関しては、取り上

げている書籍が (Hilbert流についてと同様、シーケント計

算についても)少ないので、ここにシーケント計算の完全

性も示す。

シーケント、イニシャルシーケントの定義は 2.3節と

同じ。

4.2.1 シーケント計算の推論規則

推論規則は、命題様相論理 K のシーケント計算の推論

規則から Necを取り除き、以下を加えたものである。

□Γ,Γ → □∆,□A,A

□Γ → □∆,□A
Nec45

□Γ,Γ →
□Γ → NecD45

4.2.2 シーケント計算の健全性 ·完全性

健全性は容易。以下に完全性を示す。

定理 4.1 シーケント S が証明可能でないならば、世界間

の到達可能関係が serial, transitiveおよび Euclideanであっ

てかつ S を偽とする3クリプキ ·モデル4が存在する。

証明 以下の過程では、シーケントの→の左同士あるい
は右同士に同一の論理式が複数現れた場合、暗黙のうちに

随時Weakを逆向きに適用し、→の左同士あるいは右同
士に同一の論理式が複数現れないように保つものとする。

S = S0と置き、S0に対し、⊃左,⊃右,¬左,¬右を可能
な限り逆向きに適用すると、少なくとも 1つ、p1, · · · , pk,
□A1, · · · ,□Am → q1, · · · , qℓ,□B1, · · · ,□Bn の形 (ここ

で k ≥ 0, m ≥ 0, ℓ ≥ 0, n ≥ 0であり、p1, · · · , pk, q1, · · · ,
qℓ は原子命題) の、証明可能でないシーケントが得られ

る。これを S′
0 とする。

S′ = S′
0 と置いて、以下を繰り返す。

• n > 0の場合、S′ にWeakと Nec45を逆向きに適

用して、n個のシーケント□A1, · · · ,□Am, A1, · · · ,
Am → □B1, · · · ,□Bn, Bi (1 ≤ i ≤ n)を得る。これ

らを S1, · · · , Snとする。また、n = 0の場合は、S′

にWeakとNecD45を逆向きに適用し、1個のシー

ケント □A1, · · · ,□Am, A1, · · · , Am →を得て、こ
れを S1とする。これらはいずれも証明可能でない。

• 各 Si (n > 0の場合は 1 ≤ i ≤ n、n = 0の場合は

i = 1)に対し、⊃左, ⊃右, ¬左, ¬右を可能な限り
逆向きに適用すると、i 毎に少なくとも 1つ、pi1,

· · · , piki
,□A1, · · · ,□Am,□Ai

1, · · · ,□Ai
mi

→ qi1,

· · · , qiℓi ,□B1, · · · ,□Bn,□Bi
1, · · · ,□Bi

ni
の形 (ここ

で ki ≥ 0, mi ≥ 0, ℓi ≥ 0, ni ≥ 0であり、pや qに

添字の付いたものは原子命題)の、証明可能でない

シーケントが得られる。これら (i毎に 1つ選ぶ)を

S′
iとする。S′

iの→の左右の、□Aの形のシーケン

トは、S′ に比べて減ることはないことに注意。

3シーケント A1, · · · , Am → B1, · · · , Bn の真理値は、論理式 A1,
· · · , Am が全て真で B1, · · · , Bn が全て偽のときに限り偽、それ以外の
場合は真と定義する。これは 2.3.3節での定義と同等である。

4クリプキ ·モデルの定義は [2] による。

• ここで、mi > 0または ni > 0となるような iが 1

つでもある場合は、そのような iのうち 1つを選び、

S′
iを改めて S′と置いて (従ってmや nもそれぞれ

mi, ni ほど増える)、繰り返しの先頭に戻る。さも

なければ繰り返しを終了。

この繰り返しは有限回で終了する。すなわち、→の左右
の □Aの形のシーケントがいつまでも増え続けることは

ない。なぜなら、→の左右に来うる論理式は、S に現れ

る論理式の部分論理式に限られ、それは有限通りしかな

いためである。

続いて以下のようなクリプキ ·モデルM = ⟨W,R, V ⟩
を作る。すると、Rは serial, transitive, Euclideanである。

S′
i は証明可能でないので、S′

i の→の左右に同じ原子命
題が現れることはない。

• W = {w0, w1, · · · , wn} (n > 0の場合)またはW =

{w0, w1} (n = 0の場合)

• R = {⟨wi, wj⟩ | j > 0} すなわち、どの世界から

も w0 以外の全ての世界が到達可能

• V (wi, p) =


⊤(真) if S′

i の→の左に pが現れる

⊥(偽) if S′
i の→の右に pが現れる

どちらでもいい if 上記以外

すると、各 i (i ≥ 0)について、Si から S′
i を導く過程で

経由する各シーケント (Si, S′
iも含む)の→の左に現れる

論理式はモデルM の世界 wiで真、→の右に現れる論理
式は wiで偽となる。このことは、論理式の構造に関する

帰納法で以下のように示せる。

• 原子命題については、V (wi, p)の作り方から明らか。

• →の左の A ⊃ B の形の論理式は、S′
i に至るまで

にどこかで⊃左を適用され、→の右の Aか→の
左のBのどちらか一方が、S′

iに至る過程に現れる。

前者が現れる場合は、帰納法の過程からAが wiで

偽、従ってA ⊃ Bはwiで真。後者が現れる場合も

同様。

• →の右の A ⊃ B、→の左右の ¬Aの形の論理式に
ついては上と同様。

• →の左の □Aの形の論理式は、S′ の→の左にも
現れているため、各 Sj (j > 0)の→の左に Aが現

れている。帰納法の仮定から各 wj で Aが真、よっ

て wi で □Aは真。

• →の右の □Aの形の論理式は、S′ の→の右にも
現れているため、ある Sj (j > 0)の→の右に Aが

現れている。帰納法の仮定から wj で Aが偽、よっ

て wi で □Aは偽。

6



従って S0 は (すなわち S は) w0 で偽である。

系 4.2 KD45のシーケント計算は完全である。

証明 論理式Aが証明可能でないならば、シーケント→ A

も証明可能でなく、ゆえに→ Aを (よって Aを)偽とす

るクリプキ ·モデルが作れる。

以上からは、KD45についても、シーケントを (あるい

は論理式を)与えて、それが証明可能ならその証明を実際

に与え、さもなければ証明可能でないと答えるアルゴリ

ズムが存在することがわかる。

4.3 体系 KD45でのHilbert 流の演繹体系の完
全性

今度も、Hilbert流の公理のうち適当に選んだ 1つを T

とする。正規演繹の定義も 3.3節と同じである。

定理 4.3 任意の証明可能なシーケント S に対し、S の

Hilbert流正規演繹が存在する。

証明 定理 3.2とほぼ同じ。特に、

• Nec45 の前提に Hilbert 流正規演繹が存在する

(□A1, · · · ,□Am, A1, · · · , Am,¬□B1, · · · ,¬□Bn,

¬□B,¬B ⊢◦ ¬T ) 場合、演繹定理と PD2 から

⊢◦ □A1 ⊃ · · · ⊃ □Am ⊃ A1 ⊃ · · · ⊃ Am ⊃
¬□B1 ⊃ · · · ⊃ ¬□Bn ⊃ ¬□B ⊃ B となり、Nec

より ⊢◦ □(□A1 ⊃ · · · ⊃ □Am ⊃ A1 ⊃ · · · ⊃ Am ⊃
¬□B1 ⊃ · · · ⊃ ¬□Bn ⊃ ¬□B ⊃ B)を得る。

公理 K と MP の適用を繰り返して □□A1, · · · ,
□□Am,□A1, · · · ,□Am,□¬□B1, · · · ,□¬□Bn,

□¬□B ⊢◦ □B となるので、公理 4 と 5 およ

び IN より □A1, · · · ,□Am,¬□B1, · · · ,¬□Bn,

¬□B ⊢◦ ¬T となり、Nec45の結論にも Hilbert流

正規演繹が存在する。

• NecD45 の前提に Hilbert 流正規演繹が存在する

(□A1, · · · ,□Am, A1, · · · , Am ⊢◦ ¬T )場合、演繹定

理から ⊢◦ □A1 ⊃ · · · ⊃ □Am ⊃ A1 ⊃ · · · ⊃ Am ⊃
¬T となり、Necより ⊢◦ □(□A1 ⊃ · · · ⊃ □Am ⊃
A1 ⊃ · · · ⊃ Am ⊃ ¬T )を得る。
公理 K と MP の適用を繰り返して □□A1, · · · ,
□□Am,□A1, · · · ,□Am ⊢◦ □¬T となり、公理 4

より □A1, · · · ,□Am ⊢◦ □¬T を得る。一方、Nec

と D から ⊢◦ ¬□¬T なので、IN より □A1, · · · ,
□Am ⊢◦ ¬T となり、NecD45 の結論にも Hilbert

流正規演繹が存在する。

系 4.4 KD45の Hilbert流の演繹体系は健全かつ完全。

証明 系 3.3と同様。

また、KD45に対しても、恒真な論理式Aに対し、Aの

Hilbert流体系での証明を具体的に作れることがわかる。

5 おわりに

シーケント計算での証明から Hilbert流の演繹への変

換によって、命題非様相論理や命題様相論理 K, KD45の

Hilbert流の演繹体系の完全性が示せること、および同体

系での証明を作り出すアルゴリズムが構成できることを

述べた。同様の手法は、シーケント計算による演繹体系

を持つものなら、他の論理体系にも応用できると考えら

れる。

参考文献

[1] 萩谷昌己,西崎真也. 論理と計算のしくみ. 岩波書店,

2007.

[2] 加藤暢,高田司郎,新出尚之. 数理論理学—合理的エー

ジェントへの応用に向けて—. コロナ社, 2014.

A 付録

A.1 Hilbert 流の体系での証明作成プログラム

本稿では、Kあるいは KD45の Hilbert流の演繹体系に

対し、論理式を与えてその証明を作り出すアルゴリズム

の存在についても述べた。

実際にそのアルゴリズムを Prologで実装したプログラ

ム (SWI-Prologで動作する)のファイルが、本PDFファイ

ルに添付されており、Adobe Readerのような添付ファイ

ルを閲覧する機能を持った PDFリーダを用いれば取り出

すことができるほか、UNIX 系 OSの場合は PDFtkとい

うソフトウェアをインストールすれば、以下の手順で取

り出すこともできる。

$ pdftk このPDFファイル unpack_files �

なお、取り出されるファイル名は hilbert.plとなっている。

既存の hilbert.plというファイルがある場合、誤って上書

きしないよう注意されたい。

追記 このプログラムの以前の版では、様相オペレータ

□を、空リストと同じ記号 []で表現していた。しかし、

SWI-Prologのバージョン 6以降では、空リストの記号 []

を op述語でオペレータとして用いることができなくなっ

たため、本プログラムは正常に動作しなくなっていた。

本プログラムの現在の版は、様相オペレータ□を #で

表現 (これなら何とか四角形に見える)するよう改められ

たため、この問題は起きなくなっている。

A.2 ∧, ∨を略記としない場合の演繹体系

∧や ∨を略記とせず、正式な論理オペレータとする場
合、命題 (非様相)論理のHilbert流の健全かつ完全な演繹

体系は、2.1節で示したものに、以下の公理を加えれば得

られる。

Ax4 : A ∧B ⊃ A

Ax5 : A ∧B ⊃ B

Ax6 : A ⊃ (B ⊃ A ∧B)

Ax7 : A ⊃ A ∨B

Ax8 : B ⊃ A ∨B

Ax9 : (A ⊃ C) ⊃ ((B ⊃ C) ⊃ (A ∨B ⊃ C))

また、これらの公理を 3.1節や 4.1節で示した体系に追

加すれば、命題様相論理 Kや KD45のHilbert流の健全か

つ完全な演繹体系も得られる。健全性や完全性の証明も、

本稿本文と同様に可能である。

A.3 公理の別な選び方

文献によっては、2.1.1節の公理のうち公理 Ax3を省い

て、代わりに以下の公理を置いているものもある。

Ax3’ : (¬A ⊃ ¬B) ⊃ (B ⊃ A)

このように変更した演繹体系でも、健全性・完全性はと

もに成り立つ。

完全性が成り立つことを示すには、この (変更後の)体

系で Ax3が定理であることを示せば十分である。まず、

Weakと演繹定理はその証明に Ax3を用いていないため、

この体系でも使えることに注意。また、以下の補助規則

が使える。

¬A ⊃ ¬B
B ⊃ A

CT

これは Ax3’とMPからすぐ出る。そこで、T を任意に選

んだ公理とすると

¬A ¬A ⊃ B
B

MP

¬A ¬A ⊃ ¬B
¬B MP

¬¬T ⊃ ¬B Weak

B ⊃ ¬T CT

¬T MP

より ¬A ⊃ ¬B,¬A ⊃ B,¬A ⊢ ¬T であり、演繹定理か
ら ¬A ⊃ ¬B,¬A ⊃ B ⊢ ¬A ⊃ ¬T となる。T が公理で

あることから CT, MPにより ¬A ⊃ ¬B,¬A ⊃ B ⊢ Aと

なり、再び演繹定理によって Ax3が出る。

一方、本稿本文 (2.1節)の体系で Ax3’が定理であるこ

とは (完全性からも従うが)、2.4.1節の PD2から演繹定理

で直ちにわかる。このようなわけで、Ax3と Ax3’はいず

れを公理として採用してもよい。

/* -*-prolog-*- */

/*
   命題様相論理K, KD45のHilbert流演繹体系での証明を作るPrologプログラム
   SWI-Prologで動作する(他の処理系でも動くかも)
				     by NIDE, N. nide@ics.nara-wu.ac.jp
   Ver. 2014-06-20(第1版)
     初版
   Ver. 2016-01-12(第2版)
     従来、必然オペレータ□を表すのに空リストと同じ記号[]を使っていたが、
     SWI-Prologのver.6以降で[]をオペレータにできなくなったため、その旨の
     説明を追加するとともに、オペレータでなく空リストとして使われている
     []の表記を「[ ]」に変更した(エディタで[]を別な記号に一斉置換するこ
     とで、SWI-Prologのver.6以降での使用にも対応できるようにするため)
   Ver. 2017-09-10(第3版)
     上記と同じ理由で、オペレータとして使われていた「[]」を「#」に変更。
     これにより、エディタでの一斉置換などよる変更を加えることなく、SWI-
     Prologのver.6以降でもそのまま使用できる

   使い方:

	?- deduction(論理式, Tree).
	論理式の証明木を作成してTreeに入れる

	?- prove(論理式).
	論理式の証明木を作成して出力する

	   以上いずれも、デフォルトでは体系Kでの証明を行う
	   もちろんその論理式が証明できないならfalseになる
	   論理式には/\, \/, =>, ~, #, <>の各オペレータが使える(それぞれ
	   ∧∨⊃¬□◇を表す)
	   但し、オペレータが連続する場合は離して書くこと(1つの記号と見な
	   されるのを防ぐため) 例えば<><>pではなく<> <>pと書く
	   従来、必然オペレータ□を表すのに空リストと同じ記号[]を使ってい
	   たが、上述の理由で#に変更されたので注意

	?- set_system_type(k). あるいは ?- set_system_type(kd45).
	体系をKあるいはKD45に切り替える(次に明示的に切り替えるまではその
	体系での証明を行う)

   例:
	?- prove(#(p/\q)=> #p).
	 … Kでの□(p∧q)⊃□pの証明木を出力する …
	true.
	?- prove(#p=> #(p/\q)).
	false.
	?- set_system_type(kd45), prove(#(p/\q)=> # <>p).
	 … KD45での□(p∧q)⊃□◇pの証明木を出力する …
	true.
	?- prove(<> <>p=> <>(p\/q)).
	 … KD45に切り替え済のため、KD45での◇◇p⊃◇(p∨q)の証明木を出力 …
	true.

   但し/\, \/, <>は略記であり、=>, ~, #による正式な形に直されてから
   seq_deduction/2 (シーケントのHilbert流演繹を作り出す述語)に渡される
   従ってseq_deduction述語を直接呼ぶ場合は論理式は=>, ~, #だけで記述
   せねばならない

   prove/1が証明木を出力する場合は以下のフォーマットで出力される

   例:
	?- prove(#(p/\q)=> #p).
	1 [] ax1: p=> ~ ~q=>p
	2 [] ax1: (p=> ~ ~q=>p)=> ~p=>p=> ~ ~q=>p
	3 [1, 2] mp: ~p=>p=> ~ ~q=>p
	 …
	37 [27, 36] mp: ~ (p=> ~q)=>p
	38 [37] nec: # (~ (p=> ~q)=>p)
	39 [] k: # (~ (p=> ~q)=>p)=> # ~ (p=> ~q)=> #p
	40 [38, 39] mp: # ~ (p=> ~q)=> #p
	true.

   一番左はステップ数、その次は前提にした論理式のリスト(よって公理の場合は
   空リスト)、その次は公理や規則の名前(ax1やkは公理、mpは三段論法、necは
   必然化)。その次の「:」の右が論理式で、最後の論理式 # ~(p=> ~q)=> #p は
   proveの引数の略記を正式な形に直したものとなる

   最短の証明が求まるわけではないことに注意。上の例でも途中の ~(p=> ~q)=>p
   を単独で証明させると上の例よりはるかに長いステップ数を要する

   また、特段の高速化もしていない

   内部では、演繹の木をproof_tree(A, Rule, ListOfTree)という再帰的な項で
   表現。第1引数は論理式、第2引数は公理や規則の名前あるいは「hypo」(仮定の
   場合)、第3引数は子ノードのリスト。例えば、仮定pとp=>qからqを導く演繹木は

	proof_tree(q, mp, [
		proof_tree(p, hypo, [ ]),
		proof_tree(p=>q, hypo, [ ])
	])

   のように表す。葉ノード(公理あるいは仮定)は第3引数が空リストとなる。
   葉ノードのうち第2引数を「hypo」とするものが存在しなければ、その木は
   証明の木ということになる
*/

/* オペレータ宣言 */
:-	op(497, yfx, /\), % ∧
	op(498, yfx, \/), % ∨
	op(499, xfy, =>), % ⊃
	op(496, fy, ~),   % ¬
	op(496, fy, #),  % □
	op(496, fy, <>).  % ◇

/* 第1引数に与えた論理式の略記を展開したものを第2引数に入れる */
expand_abbrev(X, X) :-
	atom(X), !.
expand_abbrev(X/\Y, ~(X1=> ~Y1)) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(X\/Y, ~X1=>Y1) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(X=>Y, X1=>Y1) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(~X, ~X1) :-
	expand_abbrev(X, X1).
expand_abbrev(#X, #X1) :-
	expand_abbrev(X, X1).
expand_abbrev(<>X, ~ # ~X1) :-
	expand_abbrev(X, X1).

/*  公理の証明木の作成。第1引数に結論、第2引数にその証明木を入れる */
make_ax1(AX, proof_tree(AX, ax1, [ ])) :-
	AX = A=>_=>A.
make_ax2(AX, proof_tree(AX, ax2, [ ])) :-
	AX = (A=>B=>C)=>(A=>B)=>A=>C.
make_ax3(AX, proof_tree(AX, ax3, [ ])) :-
	AX = (~A=> ~B)=>(~A=> B)=>A.
make_k(AX, proof_tree(AX, k, [ ])) :-
	AX = #(A=>B)=>(#A=> #B).
make_d(AX, proof_tree(AX, d, [ ])) :- system_type(kd45),
	AX = #A=> ~ # ~A.
make_4(AX, proof_tree(AX, 4, [ ])) :- system_type(kd45),
	AX = #A=> # #A.
make_5(AX, proof_tree(AX, 5, [ ])) :- system_type(kd45),
	AX = ~ #A=> # ~ #A.
/* 推論規則の適用。第1引数に結論、第2引数に前提の証明木のリストを与え、
   作った木を第3引数に入れる */
make_mp(B, TreeList, NewTree) :-
	TreeList = [proof_tree(A, _, _), proof_tree(A=>B, _, _)],
	NewTree = proof_tree(B, mp, TreeList).
make_nec(#A, TreeList, NewTree) :-
	TreeList = [proof_tree(A, _, _)],
	NewTree = proof_tree(#A, nec, TreeList).
/* 各補助規則の適用(引数の順序はmake_mpと同じ)、
   あるいは各補助定理の証明木の作成(引数の順序はmake_ax1等と同じ) */
make_weak(A, [Tree1], NewTree) :-
	make_ax1(_, NewTree1),
	make_mp(A, [Tree1, NewTree1], NewTree).
make_mp2(A, [Tree1, Tree2], NewTree) :-
	make_ax2(_, NewTree1),
	make_mp(_, [Tree2, NewTree1], NewTree2),
	make_mp(A, [Tree1, NewTree2], NewTree).
make_mp1(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree2], NewTree1),
	make_mp2(A, [Tree1, NewTree1], NewTree).
make_mp3(A, [Tree1, Tree2], NewTree) :- % BとA⊃(B⊃C)からA⊃C 演繹定理で使用
	make_weak(_, [Tree1], NewTree1),
	make_mp2(A, [NewTree1, Tree2], NewTree).
make_id(A, NewTree) :-
	make_ax1(_=>p=>_, NewTree1), % pの箇所は何でもOK
	make_ax1(_, NewTree2),
	make_mp2(A, [NewTree1, NewTree2], NewTree).
make_pd1(A, [Tree1, Tree2], NewTree) :-
	make_ax3(_, NewTree1),
	make_mp(_, [Tree1, NewTree1], NewTree2),
	make_mp(A, [Tree2, NewTree2], NewTree).
make_pd2(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree2], NewTree2),
	make_pd1(A, [Tree1, NewTree2], NewTree).
make_in(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree1], NewTree1),
	make_pd2(A, [NewTree1, Tree2], NewTree).
make_in_subtheo(A=>B=>C, NewTree) :- % INから得る補助定理
	make_in(C, [
		proof_tree(A, hypo, [ ]),
		proof_tree(B, hypo, [ ])
	], NewTree1), % AとBからCを演繹(実際にはA=¬B)
	deduction_theorem(NewTree1, B, NewTree2), % ¬BからB⊃Cを演繹
	deduction_theorem(NewTree2, A, NewTree). % ¬B⊃B⊃Cを証明
make_dn(A, [Tree1], NewTree) :-
	make_weak(_, [Tree1], NewTree1),
	make_id(_, NewTree2),
	make_pd1(A, [NewTree1, NewTree2], NewTree).
make_dn_subtheo(A=>B, NewTree) :- % DNから得る補助定理
	make_dn(B, [
		proof_tree(A, hypo, [ ])
	], NewTree1), % AからBを演繹(実際にはA=¬¬B)
	deduction_theorem(NewTree1, A, NewTree). % ¬¬B⊃Bを証明

/* 演繹定理の適用。Γ,AからのBの演繹(様相論理の場合は正規演繹、以降でも同様)を
   第1引数に、Aを第2引数に与え、ΓからのA⊃Bの演繹を作って第3引数に入れる */
deduction_theorem(Tree, A, NewTree) :-
	reduce_hypo(Tree, A, NewTree1),
	(nonvar(NewTree1) ->
		NewTree = NewTree1 % A⊃Bの演繹を作った
	;
		make_weak(A=>_, [Tree], NewTree2), % A⊃Bの演繹を作った
		opt_tree(NewTree2, NewTree)
	).
/* deduction_theoremに似るが、仮定にAがない場合第3引数を未代入のままにする */
reduce_hypo(proof_tree(A, hypo, [ ]), B, NewTree) :- A == B, !,
	make_id(A=>_, NewTree).
reduce_hypo(proof_tree(_, _, [ ]), _, _) :- !.
reduce_hypo(proof_tree(_, mp, [Tree1, Tree2]), A, NewTree) :- !,
	reduce_hypo(Tree1, A, NewTree1), % NewTree1はA⊃Bの演繹または未代入
	reduce_hypo(Tree2, A, NewTree2), % NewTree2はA⊃B⊃Cの演繹または未代入
	(nonvar(NewTree1) ->
		(nonvar(NewTree2) ->
			make_mp2(_, [NewTree1, NewTree2], NewTree3)
		;
			make_mp1(_, [NewTree1, Tree2], NewTree3)
		), % A⊃Cの演繹を作った
		opt_tree(NewTree3, NewTree)
	;
		(nonvar(NewTree2) ->
			make_mp3(_, [Tree1, NewTree2], NewTree3),
			opt_tree(NewTree3, NewTree) % A⊃Cの演繹を作った
		;
			NewTree = _ % 仮定にAなし
		)
	).
reduce_hypo(proof_tree(_, nec, _), _, _). % 仮定はないはず
	
/* 証明木の最適化用。第1引数に与えた証明木の部分木で、第1引数と同じ論理式を
   既に証明できている部分があれば、そこを取り出し第2引数に入れる */
opt_tree(Tree, OptTree) :-
	Tree = proof_tree(A, _, _),
	conclu_subtree(A, Tree, _, OptTree, _).
/* 第2引数の木の部分木で、第1引数の論理式を結論に持つ最小のものを第4引数に、
   そのサイズを第5引数に入れる。なければそれらを未代入のままにし、木自体の
   サイズを第3引数へ */
conclu_subtree(A, Tree, Size, MinConcl, MinConclSize) :-
	Tree = proof_tree(B, _, L),
	list_conclu_subtree(A, L, S1, M1, MS1),
	(nonvar(M1) ->
		MinConcl = M1, MinConclSize = MS1
	;
		(A == B ->
			MinConcl = Tree, MinConclSize is S1 + 1
		;
			Size is S1 + 1
		)
	).
list_conclu_subtree(_, [ ], 0, _, _) :- !.
list_conclu_subtree(A, [Tree|L], SizeSum, MinConcl, MinConclSize) :-
	list_conclu_subtree(A, L, SizeSum1, MinConcl1, MinConclSize1),
	conclu_subtree(A, Tree, Size, MinConcl2, MinConclSize2),
	(nonvar(MinConcl1) ->
		(nonvar(MinConcl2), MinConclSize1 > MinConclSize2 ->
			MinConcl = MinConcl2, MinConclSize = MinConclSize2
		;
			MinConcl = MinConcl1, MinConclSize = MinConclSize1
		)
	;
		(nonvar(MinConcl2) ->
			MinConcl = MinConcl2, MinConclSize = MinConclSize2
		;
			SizeSum is SizeSum1 + Size
		)			
	).

/* Treeが(仮定なしの)証明木になっているかどうかのチェック */
check_tree(Tree) :- ground(Tree), check_tree_aux(Tree), !.
check_tree_aux(Tree) :- make_ax1(_, Tree).
check_tree_aux(Tree) :- make_ax2(_, Tree).
check_tree_aux(Tree) :- make_ax3(_, Tree).
check_tree_aux(Tree) :- make_k(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_d(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_4(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_5(_, Tree).
check_tree_aux(Tree) :-
	make_mp(_, [Tree1, Tree2], Tree),
	check_tree_aux(Tree1),
	check_tree_aux(Tree2).
check_tree_aux(Tree) :-
	make_nec(_, [Tree1], Tree),
	check_tree_aux(Tree1).

/* 木の出力 */
output_tree(Tree) :- output_tree(Tree, 0, _).
output_tree(proof_tree(A, Rule, ChildrenTreeList), N, N1) :-
	output_treelist(ChildrenTreeList, N, N2, NL), N1 is N2+1,
	write(N1), write(' '), write_numslist(NL), write(' '), write(Rule),
	write(': '), write(A), nl.
output_treelist([ ], N, N, [ ]).
output_treelist([Tree|List], N, N1, [N2|NL]) :-
	output_tree(Tree, N, N2),
	output_treelist(List, N2, N1, NL).
write_numslist(L) :- write(L).

/* 定理(何でもよいので1つ選んで固定しておく)の証明木、あるいはその否定の作成 */
make_theorem(T, Tree) :- T = p=>p=>p, make_ax1(T, Tree).
not_theorem(~(p=>p=>p)).

/* 第1引数に与えたシーケントに対するHilbert流演繹を作成して第2引数に入れる
   第1引数はシーケントの→の左のリストと右のリストのペア
   シーケントの証明を作ってからHilbert流演繹に変換するのではなく
   シーケントの証明を作るためのWangのアルゴリズムを適用しつつ、その過程で
   Hilbert流演繹を直接作り出している
   3引数の場合の第3引数は(シーケント計算の)規則Nec, Nec45, NecD45を適用
   する直前のシーケントの→の左右それぞれから一番外の□を取り除いたリストのペア
   KD45の場合にこれが前回と同じになったらそこで規則の適用を止めることによって
   規則の適用の無限ループを防ぐためにある Kの場合は使っていない */
seq_deduction((Left, Right), NewTree) :-
	seq_deduction((Left, Right), NewTree, ([ ], [ ])).
seq_deduction((Left, Right), NewTree, _) :-
	member(A, Left), member(A, Right), !, % イニシャルシーケント
	not_theorem(NT),
	make_in(NT, [
		proof_tree(~A, hypo, [ ]), proof_tree(A, hypo, [ ])
	], NewTree). % ¬A,Aから¬Tの演繹を作った
seq_deduction((Left, Right), Tree, OrigLR) :-
	select(~A, Left, Left1), !, % →の左に¬Aあり
	% Γ,¬A,¬Δから¬Tの演繹を作る
	seq_deduction((Left1, [A|Right]), Tree, OrigLR).
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(~A, Right, Right1), !, % →の右に¬Aあり
	% Γ,A,¬Δから¬Tの演繹を作り、仮定のAを¬¬AからAの演繹に置き換え
	seq_deduction(([A|Left], Right1), Tree, OrigLR),
	make_dn(A, [proof_tree(~ ~A, hypo, [ ])], ReplTree), % ¬¬AからAの演繹
	replace_hypo(A, ReplTree, Tree, NewTree). % 仮定のAを¬¬Aに
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(A=>B, Right, Right1), !, % →の右にA⊃Bあり
	% Γ,A,¬Δ,¬Bから¬Tの演繹を作り、仮定のAと¬Bをそれぞれ
	% ¬(A⊃B)からA、¬(A⊃B)から¬Bの演繹に置き換え
	seq_deduction(([A|Left], [B|Right1]), Tree, OrigLR),
	make_weak(_, [proof_tree(~(A=>B), hypo, [ ])], ReplTree1),
	make_in_subtheo(_, ReplTree2),
	make_pd1(A, [ReplTree1, ReplTree2], ReplTree3), % ¬(A⊃B)からAの演繹
	make_weak(_, [proof_tree(~(A=>B), hypo, [ ])], ReplTree4),
	make_dn_subtheo(_, ReplTree5),
	make_ax1(_, ReplTree6),
	make_mp1(_, [ReplTree5, ReplTree6], ReplTree7),
	make_pd1(~B, [ReplTree4, ReplTree7], ReplTree8), %¬(A⊃B)から¬Bの演繹
	replace_hypo(A, ReplTree3, Tree, NewTree1), % 仮定のAを¬(A⊃B)に
	replace_hypo(~B, ReplTree8, NewTree1, NewTree). % 仮定の¬Bを¬(A⊃B)に
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(A=>B, Left, Left1), !, % →の左にA⊃Bあり
	% Γ,¬Δ,¬Aから¬Tの演繹を作り、Γ,¬Δから¬A⊃¬Tの演繹に作り替える
	seq_deduction((Left1, [A|Right]), Tree1, OrigLR),
	deduction_theorem(Tree1, ~A, ReplTree1),
	% Γ,B,¬Δから¬Tの演繹を作り、Γ,¬ΔからB⊃¬Tの演繹に作り替える
	seq_deduction(([B|Left1], Right), Tree2, OrigLR),
	deduction_theorem(Tree2, B, ReplTree2),
	% これらを用いてΓ,A⊃B,¬Δから¬Tの演繹を作る
	make_theorem(_, ReplTree3),
	make_pd2(_, [ReplTree1, ReplTree3], ReplTree4),
	make_mp(_, [ReplTree4, proof_tree(A=>B, hypo, [ ])], ReplTree5),
	make_mp(_, [ReplTree5, ReplTree2], NewTree). % Γ,A⊃B,¬Δから¬T演繹
seq_deduction((Left, Right), NewTree, OrigLR) :- system_type(k),
	nonnec(Left, Left0), sort(Left0, Left1),
	nonnec(Right, Right0), sort(Right0, Right1),
	member(B, Right1), % →の右に□Bあり ここでカットしない 
	% Γ,¬Bから¬Tの演繹を作り、Γから¬B⊃¬Tの演繹に作り替える
	seq_deduction((Left1, [B]), Tree, OrigLR), !, % OrigLRはKでは使わない
	deduction_theorem(Tree, ~B, NewTree1), % Γから¬B⊃¬Tの演繹
	make_theorem(_, NewTree2),
	make_pd2(_, [NewTree1, NewTree2], NewTree3), % ΓからBの演繹
	reverse(Left1, RevLeft1),
	repeat_deduction_theorem(RevLeft1, NewTree3, NewTree4), % Γ⊃Bの証明
	make_nec(_, [NewTree4], NewTree5),
	repeat_make_neck_mp(Left1, NewTree5, NewTree6), % □Γ⊃□Bの証明
	not_theorem(NT),
	make_in(NT, [proof_tree(~ #B, hypo, [ ]), NewTree6], NewTree).
seq_deduction((Left, Right), NewTree, (OrigL, [ ])) :- system_type(kd45),
	\+ member(#_, Right), !, % →の右に□Bなし
	nonnec1(Left, NLeft0), sort(NLeft0, NLeft1), nonnec(NLeft1, Left1),
	Left1 \= OrigL, merge_set(NLeft1, Left1, Left2), 
	% □Γ,Γから¬Tの演繹を作り、∧(□Γ∪Γ)⊃¬Tの証明に作り替える
	seq_deduction((Left2, [ ]), Tree, (Left1, [ ])),
	reverse(Left2, RevLeft2),
	repeat_deduction_theorem(RevLeft2, Tree, NewTree1),
	make_nec(_, [NewTree1], NewTree2), % □(∧(□Γ∪Γ)⊃¬T)の証明
	repeat_make_neck_mp(Left2, NewTree2, NewTree3),
	apply_45((Left1, [ ]), NewTree3, NewTree4), % □Γ⊃□¬Tの証明
	make_theorem(_, NewTree5), make_nec(_, [NewTree5], NewTree6),
	make_d(_, NewTree7), make_mp(_, [NewTree6, NewTree7], NewTree8),
	not_theorem(NT),
	make_in(NT, [NewTree8, NewTree4], NewTree). % □Γ⊃¬Tの証明
seq_deduction((Left, Right), NewTree, OrigLR) :- system_type(kd45),
	nonnec1(Left, NLeft0), sort(NLeft0, NLeft1), nonnec(NLeft1, Left1),
	nonnec1(Right, NRight0), sort(NRight0, NRight1),
	nonnec(NRight1, Right1),
	OrigLR \= (Left1, Right1), merge_set(NLeft1, Left1, Left2),
	member(B, Right1), % ここでカットしない
	% □Γ,Γ,¬□Σ,¬Bから¬Tの演繹を作り(ここでΣ=Δ∪{B})、
	% □Γ,Γ,¬□ΣからBの演繹に作り変える
	seq_deduction((Left2, [B|NRight1]), Tree, (Left1, Right1)), !,
	deduction_theorem(Tree, ~B, NewTree1), % □Γ,Γ,¬□Σから¬B⊃¬T
	make_theorem(_, NewTree2),
	make_pd2(_, [NewTree1, NewTree2], NewTree3), % □Γ,Γ,¬□ΣからB
	all_not(NRight1, NNRight1), reverse(NNRight1, RevNNRight1),
	repeat_deduction_theorem(RevNNRight1, NewTree3, NewTree4),
	reverse(Left2, RevLeft2),
	repeat_deduction_theorem(RevLeft2, NewTree4, NewTree5),
	make_nec(_, [NewTree5], NewTree6), % □(∧(□Γ∪Γ∪¬□Σ)⊃B)
	repeat_make_neck_mp(Left2, NewTree6, NewTree7),
	repeat_make_neck_mp(NNRight1, NewTree7, NewTree8),
	apply_45((Left1, Right1), NewTree8, NewTree9), % □Γ,¬□Σから□B
	make_in(_, [proof_tree(~ #B, hypo, [ ]), NewTree9], NewTree).
all_not([ ], [ ]).
all_not([A|L], [~A|L1]) :- all_not(L, L1).
nonnec([ ], [ ]).
nonnec([#A|L], [A|L1]) :- !, nonnec(L, L1).
nonnec([_|L], L1) :- nonnec(L, L1).
nonnec1([ ], [ ]).
nonnec1([#A|L], [#A|L1]) :- !, nonnec1(L, L1).
nonnec1([_|L], L1) :- nonnec1(L, L1).
repeat_deduction_theorem([ ], Tree, Tree).
repeat_deduction_theorem([A|L], Tree, NewTree) :-
	deduction_theorem(Tree, A, NewTree1),
	repeat_deduction_theorem(L, NewTree1, NewTree).
repeat_make_neck_mp([ ], Tree, Tree).
repeat_make_neck_mp([A|L], Tree, NewTree) :-
	make_k(_, NewTree1),
	make_mp(_, [Tree, NewTree1], NewTree2),
	make_mp(_, [proof_tree(#A, hypo, [ ]), NewTree2], NewTree3),
	repeat_make_neck_mp(L, NewTree3, NewTree).
apply_45((L, _), Tree, NewTree) :- Tree = proof_tree(# #A, hypo, [ ]), !,
	(member(A, L) ->
		make_4(_, NewTree1),
		make_mp(_, [proof_tree(#A, hypo, [ ]), NewTree1], NewTree)
	;
		NewTree = Tree
	).
apply_45((_, R), Tree, NewTree) :- Tree = proof_tree(# ~ #A, hypo, [ ]), !,
	(member(A, R) ->
		make_5(_, NewTree1),
		make_mp(_, [proof_tree(~ #A, hypo, [ ]), NewTree1], NewTree)
	;
		NewTree = Tree
	).
apply_45(LR, proof_tree(A, Type, List), proof_tree(A, Type, NewList)) :-
	apply_45_list(LR, List, NewList).
apply_45_list(_, [ ], [ ]) :- !.
apply_45_list(LR, [Tree|List], [Tree1|List1]) :-
	apply_45(LR, Tree, Tree1),
	apply_45_list(LR, List, List1).

/* 第3引数に与えた演繹の木のうち、第1引数の論理式が葉になっているところに
   ついて、それを第2引数の木で置き換えたものを作る */
replace_hypo(A, ReplTree, proof_tree(A, hypo, [ ]), ReplTree) :- !.
replace_hypo(_, _, Tree, Tree) :- Tree = proof_tree(_, _, [ ]), !.
replace_hypo(A, ReplTree, Tree, NewTree) :-
	Tree = proof_tree(B, Type, List),
	NewTree = proof_tree(B, Type, NewList),
	list_replace_hypo(A, ReplTree, List, NewList).
list_replace_hypo(_, _, [ ], [ ]) :- !.
list_replace_hypo(A, ReplTree, [Tree|Rest], [NewTree|NewRest]) :-
	replace_hypo(A, ReplTree, Tree, NewTree),
	list_replace_hypo(A, ReplTree, Rest, NewRest).

/* 論理式Aの証明を作る Aには∧∨◇が入っていてもよい */
deduction(A, Tree) :-
	expand_abbrev(A, B), % A中の略記を展開してBを得る
	seq_deduction(([ ], [B]), Tree1), % ¬Bから¬Tへの演繹を作る
	deduction_theorem(Tree1, ~B, Tree2), % ¬B⊃¬Tの証明を作る
	make_theorem(_, Tree3),
	make_pd2(_, [Tree2, Tree3], Tree4), % これでBの証明が完成
	opt_tree(Tree4, Tree),
	(check_tree(Tree) -> % 念のためTreeの正当性を検査
		true
	;
		write('A bug occurred while proving '), write(A), nl, fail
	).

/* 論理式Aの証明を作って出力 */
prove(A) :-
	deduction(A, Tree),
	output_tree(Tree).

set_system_type(SysType) :-
	member(SysType, [k, kd45]), !,
	abolish(system_type/1),
	assert(system_type(SysType)).
set_system_type_if_unset(SysType) :-
	(clause(system_type(_), _) -> true; set_system_type(SysType)).
:- dynamic(system_type/1).
:- set_system_type_if_unset(k).


